

## Paper 55-27

### Summarizing Data with Base SAS® PROCs

Deborah Babcock Buck, D. B. & P. Associates, Houston, TX

#### ABSTRACT

Base SAS provides a number of procedures designed to aid the SAS user in developing data summary reports. These procedures include MEANS, UNIVARIATE, FREQ, REPORT and TABULATE. Additionally, PROC FORMAT is also available to modify the appearance of data values or combine values into desired categories within PROCs without changing actual data values or creating new variables.

This presentation explores various common types of summary reports and what factors you should consider in deciding which procedure is best suited to your reporting needs. It is not meant to be a detailed tutorial on PROC step programming code for all of these procedures, but rather a guide on how to decide which procedures fits the requirements for a given report.

Summary reports generated using Base SAS PROCs are presented along with the code producing these reports. These examples are valid with SAS Versions 6-8.

#### INTRODUCTION

Base SAS has a number of procedures to help you report your data in the form you would like to see it presented. In deciding which procedure is most appropriate for your needs, you must consider a number of factors. These factors include the following.

- What is the purpose of the summary?
- What information do you need in the report?
- Do you need a specific layout or customized information?
- Do you need cross-tabulations or hierarchical groupings?
- Do you need statistics?

We will consider each of these items to see which PROC(s) will let you produce the desired report most easily. All of the procedures discussed in this paper are available in Base SAS. Output from the code given in this paper is shown in the Appendix. (Note: Tables 3 through 6 only include the last page of output generated by these procedures.)

#### WHAT IS THE PURPOSE OF THE REPORT?

Do you need a detail or a summary report? A detail report gives you at least one row of information in your report for every observation in the data set (assuming you are not using a WHERE statement in the procedure to limit which observations are processed). A summary report generally does some form of collapsing of observations so that each row in the report represents information from more than one observation.

Some of the most frequently used Base SAS procedures for reporting and summarizing data include the following.

- PRINT – Produces a detail report, although this can include summary information overall and/or within subgroups. PRINT can also produce a summary report if the data has been outputted to a summary data set through another procedure.
- MEANS/SUMMARY - Produces summary statistics, by default number of observations, mean, standard deviation, minimum, and maximum.

- UNIVARIATE - Produces the above statistics, plus a number of other statistics including median, mode, percentiles and the five lowest and highest values.
- FREQ – Generates counts and percentages for one-way and two-way tables; performs statistical tests on frequency data.
- TABULATE – Produces one, two and three-way summary tables with selected summary statistics.
- REPORT - Produces detail or summary reports with selected summary statistics; allows a great deal of flexibility in customizing reports.

One procedure that does not produce a report itself, but is often essential to producing customized reports is PROC FORMAT. Although SAS provides a multitude of SAS-defined formats, quite often a format is needed that is not available. PROC FORMAT allows you to create user-defined formats. In this study, we would like to see the GENDER and NRACE variables printed with user-defined formats. Note that it is a two-step process. First the format must be created; then the format must be assigned to the appropriate variable(s).

The following code sets up the formats for selected variables and uses PROC DATASETS to display the descriptor portion of the permanent SAS data set BTPAPER.MBP.

```
LIBNAME BTPAPER 'C:\SAS\BTPAPER';

OPTIONS LINESIZE=142 PAGESIZE=42 NODATE NONUMBER;

PROC FORMAT;
  VALUE $GENDER
    'M'='Male'
    'F'='Female';
  VALUE $RACE
    '1'='White'
    '2'='Black'
    '3'='Oriental'
    '4'='Other';
  VALUE $RCODE
    '001'='Baseline DBP > 90, decreased by 10% or more'
    '002'='Baseline DBP > 90, decreased by 5% up to 10%'
    '003'='Baseline DBP > 90, decreased by 0% up to 5%'
    '004'='Baseline DBP > 90, increased'
    '005'='Baseline DBP <= 90';
  VALUE NCENTER
    1='Dr. Green'
    2='Dr. Jones'
    3='Dr. Wright';
RUN;

DATA BTPAPER.MBP;
  SET BTPAPER.MBP;
  FORMAT NRACE $RACE. GENDER $GENDER.;
RUN;

**** CONTENTS OF ORIGINAL DATA SET ****;
**** BTPAPER.MBP (TABLE 1) ****;

PROC DATASETS LIB=BTPAPER;
  CONTENTS DATA=MBP;
  TITLE1 'TABLE 1';
  TITLE2 'CONTENTS OF BTPAPER.MBP DATA SET';
RUN;
```

Appendix Table 1 shows the six demographic and diastolic blood pressure variables present in the original data set, BTPAPER.MBP.

When the purpose of the report is primarily to produce summary reports to be used for data cleaning, such as checking for outliers or looking for missing or miscoded values, the most commonly used procedures are PROC PRINT for detail reports and PROC MEANS, PROC UNIVARIATE and PROC FREQ for summary reports. These procedures are also often used to produce preliminary or "ad hoc" reports. The following code uses PROC FREQ to produce a two-way summary tables with the frequencies of GENDER by NRACE combinations.

```
**** PROC FREQ FOR BTPAPER.MBP ****;
**** SUMMARY REPORT (TABLE 2) ****;

PROC FREQ DATA=BTPAPER.MBP;
  TABLES GENDER*NRACE;
  TITLE1 'TABLE 2';
  TITLE2
  'PROC FREQ OF BTPAPER.MBP - SUMMARY REPORT';
RUN;
```

Output from this code is shown in Appendix Table 2. Results from using PROC MEANS are shown in a later example.

PROC TABULATE and PROC REPORT are the real "work-horses" of customized report writing. Both of these procedures produce summary reports, while PROC REPORT can also generate detail reports. Examples of both of these procedures are discussed in later sections.

### WHAT DATA DO YOU NEED IN THE REPORT?

Do you need additional columns of information in the report that do not already exist as variables in the data set? An example of this might be the sum or average of two existing variables in the data set. If that information is only necessary for the report, why go through an additional DATA step to create a new variable? PROC REPORT will allow you to create new columns of information within the PROC step.

If you need the same information represented in more than one way in a report (for example, a medical code represented as a coded value in the data set as well as an interpreted or formatted value), then PROC REPORT will allow you to show both columns in the same report. PROC PRINT will also do this, but requires that you create a new variable in the DATA step. (It will not print the same variable with two different formats in the same report.) Using PROC FORMAT to create a format to associate with the coded value in conjunction with PROC REPORT will allow you to use both the coded and formatted value in the same report without creating a new variable.

The following code creates a temporary data set, WORK.MBP2 which includes the newly created variable CSTDIAM (Change in Diastolic Blood Pressure) as well as the six variables present in the original BTPAPER.MBP data set.

```
**** CREATING WORK DATA SET MBP2 ****;

DATA MBP2;
  SET BTPAPER.MBP;
  CSTDIAM=STDIAI-BSTDIAI;
  LABEL CSTDIAM='Change in Diastolic BP';
RUN;
```

Tables 3 and 4 present detail reports using PROC PRINT and PROC REPORT, respectively. Table 4 includes two additional columns of information generated with PROC REPORT. The Physician Name column is created based on the value for CENTER, and Change in Diastolic Blood Pressure is computed with a COMPUTE block. The code that generated Table 4 also creates a new data set called WORK.NEWMBP which includes

the newly computed variables. The following code produces Appendix Tables 3 and 4.

```
**** PROC PRINT FOR WORK.MBP2 ****;
**** DETAIL REPORT (TABLE 3) ****;

PROC PRINT DATA=MBP2 SPLIT='/' NOOBS;
  VAR PTNO CENTER NRACE GENDER
  BSTDIAM STDIAI CSTDIAM;
  FORMAT NRACE $RACE. GENDER $GENDER.;
  LABEL PTNO='Patient/Number'
  CENTER='Center'
  NRACE='Race'
  GENDER='Gender'
  BSTDIAM='Baseline/Diastolic/BP'
  STDIAI='Final/Diastolic/BP'
  CSTDIAM='Change in/Diastolic/BP';
  TITLE1 'TABLE 3';
  TITLE2
  'PROC PRINT OF WORK.MBP2 - DETAIL REPORT';
RUN;

**** PROC REPORT FOR BTPAPER.MBP ****;
**** DETAIL REPORT (TABLE 4) ****;
**** CREATES OUTPUT DATA SET WORK.NEWMBP ****;

PROC REPORT DATA=BTPAPER.MBP NOWD
  HEADSKIP SPACING=3 OUT=NEWMBP;
  COLUMN PTNO CENTER CENTER=NCENTER NRACE
  GENDER BSTDIAM STDIAI CSTDIAM;
  DEFINE PTNO/DISPLAY WIDTH=7 FORMAT=3.
  CENTER 'Patient Number';
  DEFINE CENTER/DISPLAY WIDTH=6 FORMAT=1.
  CENTER 'Center';
  DEFINE NCENTER/WIDTH=10 FORMAT=NCENTER.
  CENTER 'Physician Name';
  DEFINE NRACE/DISPLAY WIDTH=9 CENTER
  FORMAT=$RACE. 'Race';
  DEFINE GENDER/DISPLAY WIDTH=6 FORMAT=$GENDER.
  'Gender';
  DEFINE BSTDIAM/ANALYSIS MEAN WIDTH=9
  FORMAT=3. CENTER SPACING=5
  'Baseline/Diastolic/BP';
  DEFINE STDIAI/ANALYSIS MEAN WIDTH=9 FORMAT=3.
  CENTER 'Final/Diastolic/BP';
  DEFINE CSTDIAM/COMPUTED WIDTH=9 FORMAT=3.
  CENTER 'Change in/Diastolic/BP';
  COMPUTE CSTDIAM;
  CSTDIAM=STDIAI.MEAN-BSTDIAI.MEAN;
  ENDCOMP;
  TITLE1 'TABLE 4';
  TITLE2
  'PROC REPORT OF BTPAPER.MBP - DETAIL REPORT';
RUN;
```

### DO YOU NEED A SPECIFIC LAYOUT OR CUSTOMIZED INFORMATION?

This could include, but is not limited to the following:

- providing customized information for subgroups or at the beginning or end of a report,
- underlining column headers,
- allowing variable values to extend across more than one line.

When developing a detail report, the simplest option you should consider is PROC PRINT. If you need sums for subgroups or at the end of a report, PROC PRINT will give these to you. However, if in addition to or instead of sums, you need other statistics or information (which can include customized text), then PROC REPORT is probably the more appropriate approach. Also, if you need to underline the column headers or you have variable values that will extend past the page width (based on the number of variables you are trying to print and the PAGESIZE= option), then PROC REPORT will provide these capabilities.

Appendix Tables 5 and 6 present detail tables, although Table 6 also includes summary information. Table 5 uses PROC PRINT to produce a detail report on WORK.MBP3 (which is created in a data step below from WORK.MBP2 so that the physician name and comment variables are available for printing). PROC REPORT was used to generate Table 6. In this table the column headers are underlined, the comment variable values can extend across multiple lines, and customized text with summary statistics are provided overall and for each center. The following code produces these tables.

```

**** CREATING DATA SET WORK.MBP3 ****;
**** AND PROC PRINT OF WORK.MBP3 ****;
**** DETAIL REPORT (TABLE 5) ****;

DATA MBP3; SET MBP2;
  LENGTH COMMENT $ 3;
  NCENTER=CENTER;
  RATIO=CSTDIA/BSTDIA;
  IF BSTDIA GT 90 THEN DO;
    IF RATIO NE . AND RATIO LE -.10
      THEN COMMENT='001';
    ELSE IF -.10 LT RATIO LE -.05
      THEN COMMENT='002';
    ELSE IF -.05 LT RATIO LE 0
      THEN COMMENT='003';
    ELSE IF RATIO GT 0 THEN COMMENT='004';
  END;
  ELSE IF BSTDIA LE 90 THEN COMMENT='005';
  FORMAT NCENTER NCENTER. COMMENT $RCODE.;
RUN;

PROC SORT DATA=MBP3;
  BY CENTER NCENTER;
RUN;

PROC PRINT DATA=MBP3 SPLIT='/' NOOBS;
  ID CENTER NCENTER;
  BY CENTER NCENTER;
  VAR PTNO NRACE GENDER BSTDIA
    STDIA CSTDIA COMMENT;
  FORMAT NRACE $RACE. GENDER $GENDER.;
  LABEL PTNO='Patient/Number'
    CENTER='Center'
    NRACE='Race'
    GENDER='Gender'
    COMMENT='Comment'
    NCENTER='Physician/Name'
    BSTDIA='Baseline/Diastolic/BP'
    STDIA='Final/Diastolic/BP'
    CSTDIA='Change in/Diastolic/BP';
  TITLE1 'TABLE 5';
  TITLE2
    'PROC PRINT OF WORK.MBP3 - DETAIL REPORT';
RUN;

**** PROC REPORT FOR BTPAPER.MBP ****;
**** WITH SUMMARY INFORMATION ****;
**** DETAIL REPORT (TABLE 6) ****;

PROC REPORT DATA=BTPAPER.MBP NOWD
  HEADSKIP HEADLINE SPACING=3;
  COLUMN CENTER CENTER=NCENTER PTNO NRACE GENDER
    BSTDIA STDIA CSTDIA COMMENT;
  DEFINE PTNO/DISPLAY WIDTH=7 FORMAT=3.
    CENTER 'Patient Number';
  DEFINE CENTER/ORDER WIDTH=6 FORMAT=1.
    CENTER 'Center';
  DEFINE NCENTER/ORDER WIDTH=10 FORMAT=NCENTER.
    'Physician Name';
  DEFINE NRACE/DISPLAY WIDTH=9 CENTER
    FORMAT=$RACE. 'Race';
  DEFINE GENDER/DISPLAY WIDTH=6 FORMAT=$GENDER.
    'Gender';
  DEFINE BSTDIA/ANALYSIS MEAN WIDTH=9 FORMAT=3.
    CENTER SPACING=5 'Baseline/Diastolic/BP';
  DEFINE STDIA/ANALYSIS MEAN WIDTH=9 FORMAT=3.
    CENTER 'Final/Diastolic/BP';
  DEFINE CSTDIA/COMPUTED WIDTH=9 FORMAT=3.
    CENTER 'Change in/Diastolic/BP';

```

```

  DEFINE COMMENT/COMPUTED WIDTH=25 SPACING=5
    FLOW FORMAT=$RCODE. 'Comment';

  COMPUTE CSTDIA;
    CSTDIA=STDIA.MEAN-BSTDIA.MEAN;
  ENDCOMP;

  COMPUTE COMMENT/CHARACTER;
  IF BSTDIA.MEAN GT 90 AND
    (CSTDIA/BSTDIA.MEAN) LE -.10
    THEN COMMENT='001';
  ELSE IF BSTDIA.MEAN GT 90 AND
    -.10 LT (CSTDIA/BSTDIA.MEAN) LE -.05
    THEN COMMENT='002';
  ELSE IF BSTDIA.MEAN GT 90 AND
    -.05 LT (CSTDIA/BSTDIA.MEAN) LE 0
    THEN COMMENT='003';
  ELSE IF BSTDIA.MEAN GT 90 AND
    (CSTDIA/BSTDIA.MEAN) GT 0
    THEN COMMENT='004';
  ELSE IF BSTDIA.MEAN LE 90 THEN COMMENT='005';
  ENDCOMP;

  BREAK AFTER CENTER/SKIP;
  COMPUTE AFTER CENTER;
  PCHANGE=(CSTDIA/BSTDIA.MEAN)*100;
  LINE @3 ' ';
  LINE @22 111*'-';
  LINE @22
    '|The average change in diastolic blood'
    @61 'pressure for'
    @74 CENTER NCENTER. @85 'was'
    @89 CSTDIA 6.2
    @96 'with an average % change of'
    @124 PCHANGE 7.2 @132 '|';
  LINE @22 111*'-';
  LINE @3 ' ';
  ENDCOMP;

  RBREAK AFTER/SKIP;
  COMPUTE AFTER;
  PCHANGE=(CSTDIA/BSTDIA.MEAN)*100;
  LINE @3 ' ';
  LINE @22 103*'=';
  LINE @22
    '|The average change in diastolic blood'
    @61 'pressure overall'
    @78 'was' @82 CSTDIA 6.2
    @89 'with an average % change of'
    @116 PCHANGE 7.2 @124 '|';
  LINE @22 103*'=';
  LINE @3 ' ';
  ENDCOMP;
  TITLE1 'TABLE 6';
  TITLE2
    'PROC REPORT OF BTPAPER.MBP - DETAIL REPORT';
RUN;

```

## DO YOU NEED CROSS-TABULATIONS OR HIERARCHICAL GROUPINGS?

Both of these types of tables are summary reports where you display some sort of statistic(s) within a combination of variable values in either the row and/or column dimensions. As we have seen earlier, PROC FREQ can produce cross-tab tables. PROC PRINT can also produce this type of report after outputting the summary information from some other procedure. However, this type of report is usually generated with PROC TABULATE or PROC REPORT because of the greater flexibility and need for fewer steps.

Appendix Tables 8, 9 and 10 all show the diastolic blood pressure summarized by demographic data. Table 8 provides the same information as the other tables, but may not be as desirable for reporting purposes. It also requires that the data be outputted with PROC MEANS or SUMMARY as shown in Table 7. Tables 9 and 10 produce very similar reports. Table 9 (PROC TABULATE) requires less programming code, while Table 10 (PROC REPORT) allows more flexibility in spacing between columns of information,

underlining of headers, and customized text (if desired). The following code generates Appendix Tables 7, 8, 9 and 10.

```

**** PROC MEANS FOR WORK.MBP2 ****;
**** TO CREATE NEWMBP2 ****;
**** SUMMARY REPORT (TABLE 7) ****;

PROC MEANS DATA=MBP2 MAXDEC=2 /*NOPRINT*/;
CLASS GENDER NRACE;
VAR BSTDIAM STDIAM CSTDIAM;
OUTPUT OUT=NEWMBP2 N=BSTDIAM STDIAM CSTDIAM
      MEAN=MEANBST MEANST CMEANST
      STD=STDBST STDST CSTDST;

TITLE1 'TABLE 7';
TITLE2
  'PROC MEANS OF WORK.MBP2';
TITLE3 'SUMMARY REPORT';
RUN;

PROC PRINT DATA=NEWMBP2 LABEL NOOBS;
VAR GENDER NRACE _TYPE_ BSTDIAM MEANBST STDBST
  STDIAM MEANST STDST CSTDIAM CMEANST CSTDST;
FORMAT MEANBST MEANST CMEANST
  STDBST STDST CSTDST 8.2;
LABEL BSTDIAM='Baseline Diastolic BP - N'
  STDIAM='Final Diastolic BP - N'
  CSTDIAM='Change in Diastolic BP - N'
  MEANBST='Baseline Diastolic BP - Mean'
  MEANST='Final Diastolic BP - Mean'
  CMEANST='Change in Diastolic BP-Mean'
  STDBST='Baseline Diastolic BP - Std. Dev.'
  STDST='Final Diastolic BP - Std. Dev.'
  CSTDST='Change in Diastolic BP - Std. Dev.';
TITLE1 'TABLE 8';
TITLE2
  'PROC PRINT OF WORK.NEWMBP2';
TITLE3 'SUMMARY REPORT';
RUN;

**** PROC TABULATE FOR WORK.MBP2 ****;
**** SUMMARY REPORT (TABLE 9) ****;

OPTIONS MISSING='0';

PROC TABULATE DATA=MBP2 NOSEPS;
CLASS GENDER NRACE;
VAR BSTDIAM STDIAM CSTDIAM;
TABLE (NRACE ALL),GENDER*
  (BSTDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2)
  STDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2)
  CSTDIAM*(N*F=3. MEAN*F=6.2 STD*F=6.2))
  /RTS=15;
KEYLABEL ALL='All'
  N='No.'
  MEAN='Mean'
  STD='Std Dev.';
LABEL BSTDIAM='Baseline Diastolic BP'
  STDIAM='Final Diastolic BP'
  CSTDIAM='Change in Diastolic BP';
FORMAT NRACE $RACE. GENDER $GENDER.;
TITLE1 'TABLE 9';
TITLE2
  'PROC TABULATE OUTPUT OF WORK.MBP2';
TITLE3 'SUMMARY REPORT';
RUN;

**** PROC REPORT FOR WORK.NEWMBP2 ****;
**** SUMMARY REPORT (TABLE 10) ****;

PROC REPORT DATA=NEWMBP2 NOWD HEADSKIP
  HEADLINE SPACING=1;
COLUMN NRACE (GENDER,
  (('Baseline/Diastolic BP'
  BSTDIAM BSTDIAM-MEANBST BSTDIAM-STDBST)
  ('Final/Diastolic BP'
  STDIAM STDIAM-MEANST STDIAM-STDST)
  ('Change in/Diastolic BP'
  CSTDIAM CSTDIAM-CMEANST CSTDIAM-CSTDST)));
DEFINE NRACE/GROUP ORDER=INTERNAL WIDTH=6
  FORMAT=$RACE.;
DEFINE GENDER/ACROSS FORMAT=$GENDER.

```

```

      SPACING=2 'Gender';
DEFINE BSTDIAM/ANALYSIS N WIDTH=3 FORMAT=3.
  CENTER SPACING=3 'No.';
DEFINE MEANBST/ANALYSIS MEAN WIDTH=8
  FORMAT=8.2 CENTER 'Mean';
DEFINE STDBST/ANALYSIS STD WIDTH=6
  FORMAT=6.2 CENTER 'Std. Dev.';
DEFINE STDIAM/N WIDTH=3 FORMAT=3.
  CENTER SPACING=3 'No.';
DEFINE MEANST/MEAN WIDTH=8.2
  CENTER 'Mean';
DEFINE STDST/ STD WIDTH=6.2
  CENTER 'Std. Dev.';
DEFINE CSTDIAM/ANALYSIS N WIDTH=3 FORMAT=3.
  CENTER SPACING=3 'No.';
DEFINE CMEANST/ANALYSIS MEAN WIDTH=8
  FORMAT=8.2 CENTER 'Mean';
DEFINE CSTDST/ANALYSIS STD WIDTH=6
  FORMAT=6.2 CENTER 'Std. Dev.';

BREAK AFTER NRACE/SKIP;

RBREAK AFTER/ SKIP SUMMARIZE;

COMPUTE AFTER;
  NRACE='All';
ENDCOMP;
TITLE1 'TABLE 10';
TITLE2
  'PROC REPORT OUTPUT OF WORK.NEWMBP2';
TITLE3 'SUMMARY REPORT';
RUN;

```

## CONCLUSION

This paper has attempted to demonstrate some of the procedures available in Base SAS for generating summaries and reports. PROCs PRINT, MEANS, UNIVARIATE, FREQ, TABULATE and REPORT are just some of the many ways to produce a report using SAS software. An additional procedure in Base SAS that can be used to generate reports is PROC SQL. This procedure was not covered in this paper due to time and space limitations, but should also be considered for summarizing and reporting data. Hopefully, this paper will help guide you on which Base SAS procedure will best produce the summary you desire.

## REFERENCES

- SAS Institute Inc. (1995) *SAS Guide to the REPORT Procedure: Reference, Release 6.11*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (1990) *SAS Guide to the Report Procedure: Usage and Reference, Version 6*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (1990) *SAS Guide to TABULATE Processing, Second Edition*, Cary, NC: SAS Institute Inc.
- SAS Institute Inc. (1990) *SAS Procedures Guide, Version 6, Third Edition*, Cary, NC: SAS Institute Inc.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

## AUTHOR CONTACT INFORMATION

Debbie Buck  
 D. B. & P. Associates  
 10418 Indian Paintbrush Lane  
 Houston, TX 77095  
 Voice: 281-256-1619  
 Fax: 281-256-1634  
 Email: dbbuck@compuserve.com

## APPENDIX

TABLE 1  
CONTENTS OF BTPAPER.MBP DATA SET

DATASETS PROCEDURE

-----Alphabetic List of Variables and Attributes-----

#	Variable	Type	Len	Pos	Format	Label
1	BSTDIAM	Num	8	0		Baseline Diastolic BP
6	CENTER	Num	8	32		Center
3	GENDER	Char	3	16	\$GENDER.	Gender
4	NRACE	Char	5	19	\$RACE.	Race
5	PTNO	Num	8	24		Patient Number
2	STDIAM	Num	8	8		Final Diastolic BP

TABLE 2  
PROC FREQ OF BTPAPER.MBP - SUMMARY REPORT

TABLE OF GENDER BY NRACE

GENDER(Gender)		NRACE(Race)			Total
Frequency	Percent	White	Black	Other	
Row Pct	Col Pct				
Female	44	7	2	53	
	54.32	8.64	2.47	65.43	
	83.02	13.21	3.77		
	65.67	63.64	66.67		
Male	23	4	1	28	
	28.40	4.94	1.23	34.57	
	82.14	14.29	3.57		
	34.33	36.36	33.33		
Total	67	11	3	81	
	82.72	13.58	3.70	100.00	

TABLE 3  
PROC PRINT OF WORK.MBP2 - DETAIL REPORT

Patient Number	Center	Race	Gender	Baseline Diastolic BP	Final Diastolic BP	Change in Diastolic BP
71	3	Black	Female	103	91	-12
72	3	Black	Female	100	99	-1
73	3	Other	Female	104	103	-1
74	3	Black	Male	99	97	-2
75	3	White	Male	95	93	-2
76	3	White	Female	93	86	-7
77	3	White	Female	101	98	-3
78	3	White	Female	95	91	-4
79	3	White	Female	94	95	1
80	3	White	Female	93	84	-9
81	3	White	Female	93	95	2

TABLE 4  
PROC REPORT OF BTPAPER.MBP - DETAIL REPORT

Patient Number	Center	Physician Name	Race	Gender	Baseline Diastolic BP	Final Diastolic BP	Change in Diastolic BP
71	3	Dr. Wright	Black	Female	103	91	-12
72	3	Dr. Wright	Black	Female	100	99	-1
73	3	Dr. Wright	Other	Female	104	103	-1
74	3	Dr. Wright	Black	Male	99	97	-2
75	3	Dr. Wright	White	Male	95	93	-2
76	3	Dr. Wright	White	Female	93	86	-7
77	3	Dr. Wright	White	Female	101	98	-3
78	3	Dr. Wright	White	Female	95	91	-4
79	3	Dr. Wright	White	Female	94	95	1
80	3	Dr. Wright	White	Female	93	84	-9
81	3	Dr. Wright	White	Female	93	95	2

TABLE 5  
PROC PRINT OF WORK.MBP3 - DETAIL REPORT

Center	Physician Name	Patient Number	Race	Gender	Baseline Diastolic BP	Final Diastolic BP	Change in Diastolic BP	Comment
2	Dr. Jones	64	Black	Female	102	104	2	Baseline DBP > 90, increased
		65	White	Female	84	95	11	Baseline DBP <= 90
		66	White	Female	98	104	6	Baseline DBP > 90, increased
		67	White	Female	107	106	-1	Baseline DBP > 90, decreased by 0% up to 5%
		68	White	Female	102	104	2	Baseline DBP > 90, increased
		69	White	Female	90	92	2	Baseline DBP <= 90
3	Dr. Wright	70	White	Male	99	78	-21	Baseline DBP > 90, decreased by 10% or more
		71	Black	Female	103	91	-12	Baseline DBP > 90, decreased by 10% or more
		72	Black	Female	100	99	-1	Baseline DBP > 90, decreased by 0% up to 5%
		73	Other	Female	104	103	-1	Baseline DBP > 90, decreased by 0% up to 5%
		74	Black	Male	99	97	-2	Baseline DBP > 90, decreased by 0% up to 5%
		75	White	Male	95	93	-2	Baseline DBP > 90, decreased by 0% up to 5%
		76	White	Female	93	86	-7	Baseline DBP > 90, decreased by 5% up to 10%
		77	White	Female	101	98	-3	Baseline DBP > 90, decreased by 0% up to 5%
		78	White	Female	95	91	-4	Baseline DBP > 90, decreased by 0% up to 5%
		79	White	Female	94	95	1	Baseline DBP > 90, increased
		80	White	Female	93	84	-9	Baseline DBP > 90, decreased by 5% up to 10%
		81	White	Female	93	95	2	Baseline DBP > 90, increased

TABLE 6  
PROC REPORT OF BTPAPER.MBP - DETAIL REPORT

Center	Physician Name	Patient Number	Race	Gender	Baseline Diastolic BP	Final Diastolic BP	Change in Diastolic BP	Comment
3	Dr. Wright	70	White	Male	99	78	-21	Baseline DBP > 90, decreased by 10% or more
		71	Black	Female	103	91	-12	Baseline DBP > 90, decreased by 10% or more
		72	Black	Female	100	99	-1	Baseline DBP > 90, decreased by 0% up to 5%
		73	Other	Female	104	103	-1	Baseline DBP > 90, decreased by 0% up to 5%
		74	Black	Male	99	97	-2	Baseline DBP > 90, decreased by 0% up to 5%
		75	White	Male	95	93	-2	Baseline DBP > 90, decreased by 0% up to 5%
		76	White	Female	93	86	-7	Baseline DBP > 90, decreased by 5% up to 10%
		77	White	Female	101	98	-3	Baseline DBP > 90, decreased by 0% up to 5%
		78	White	Female	95	91	-4	Baseline DBP > 90, decreased by 0% up to 5%
		79	White	Female	94	95	1	Baseline DBP > 90, increased
		80	White	Female	93	84	-9	Baseline DBP > 90, decreased by 5% up to 10%
81	White	Female	93	95	2	Baseline DBP > 90, increased		

-----  
The average change in diastolic blood pressure for Dr. Wright was -4.92 with an average % change of -5.05

=====  
 |The average change in diastolic blood pressure overall was -2.26 with an average % change of -2.35 |  
 =====

TABLE 7  
PROC MEANS OF WORK.MBP - DETAIL REPORT

GENDER	NRACE	N Obs	Variable	Label	N	Mean	Std Dev	Minimum	Maximum
Female	White	44	BSTDIAM	Baseline Diastolic BP	44	94.70	5.52	83.00	107.00
			STDIAM	Final Diastolic BP	44	92.73	8.00	74.00	109.00
			CSTDIAM	Change in Diastolic BP	44	-1.98	6.32	-19.00	11.00
	Black	7	BSTDIAM	Baseline Diastolic BP	7	100.00	1.91	98.00	103.00
			STDIAM	Final Diastolic BP	7	99.71	4.27	91.00	104.00
			CSTDIAM	Change in Diastolic BP	7	-0.29	5.35	-12.00	3.00
	Other	2	BSTDIAM	Baseline Diastolic BP	2	101.00	4.24	98.00	104.00
			STDIAM	Final Diastolic BP	2	96.50	9.19	90.00	103.00
			CSTDIAM	Change in Diastolic BP	2	-4.50	4.95	-8.00	-1.00
Male	White	23	BSTDIAM	Baseline Diastolic BP	23	97.13	6.57	85.00	108.00
			STDIAM	Final Diastolic BP	23	93.91	9.23	73.00	110.00
			CSTDIAM	Change in Diastolic BP	23	-3.22	6.79	-21.00	10.00
	Black	4	BSTDIAM	Baseline Diastolic BP	4	98.50	5.45	94.00	106.00
			STDIAM	Final Diastolic BP	4	97.50	4.80	91.00	102.00
			CSTDIAM	Change in Diastolic BP	4	-1.00	5.60	-6.00	7.00
	Other	1	BSTDIAM	Baseline Diastolic BP	1	97.00	.	97.00	97.00
			STDIAM	Final Diastolic BP	1	90.00	.	90.00	90.00
			CSTDIAM	Change in Diastolic BP	1	-7.00	.	-7.00	-7.00

TABLE 8  
PROC PRINT OF WORK.NEWMBP2  
SUMMARY REPORT

Gender	Race	_TYPE_	Baseline	Baseline	Baseline	Final	Final	Final	Change in	Change in	Change in
			Diastolic BP - N	Diastolic BP - Mean	Diastolic BP - Std. Dev.	Diastolic BP - N	Diastolic BP - Mean	Diastolic BP - Std. Dev.	Diastolic BP - N	Diastolic BP-Mean	Diastolic BP - Std. Dev.
		0	81	96.22	5.79	81	93.96	8.11	81	-2.26	6.24
	White	1	67	95.54	5.97	67	93.13	8.39	67	-2.40	6.46
	Black	1	11	99.45	3.42	11	98.91	4.37	11	-0.55	5.16
	Other	1	3	99.67	3.79	3	94.33	7.51	3	-5.33	3.79
Female		2	53	95.64	5.51	53	93.79	7.91	53	-1.85	6.12
Male		2	28	97.32	6.22	28	94.29	8.62	28	-3.04	6.50
Female	White	3	44	94.70	5.52	44	92.73	8.00	44	-1.98	6.32
Female	Black	3	7	100.00	1.91	7	99.71	4.27	7	-0.29	5.35
Female	Other	3	2	101.00	4.24	2	96.50	9.19	2	-4.50	4.95
Male	White	3	23	97.13	6.57	23	93.91	9.23	23	-3.22	6.79
Male	Black	3	4	98.50	5.45	4	97.50	4.80	4	-1.00	5.60
Male	Other	3	1	97.00	.	1	90.00	.	1	-7.00	.

TABLE 9  
PROC TABULATE OUTPUT OF WORK.MBP2  
SUMMARY REPORT

	Gender																	
	Female									Male								
	Baseline Diastolic BP			Final Diastolic BP			Change in Diastolic BP			Baseline Diastolic BP			Final Diastolic BP			Change in Diastolic BP		
	No.	Mean	Std Dev.	No.	Mean	Std Dev.	No.	Mean	Std Dev.	No.	Mean	Std Dev.	No.	Mean	Std Dev.	No.	Mean	Std Dev.
Race	44	94.70	5.52	44	92.73	8.00	44	-1.98	6.32	23	97.13	6.57	23	93.91	9.23	23	-3.22	6.79
White	7	100.00	1.91	7	99.71	4.27	7	-0.29	5.35	4	98.50	5.45	4	97.50	4.80	4	-1.00	5.60
Black	2	101.00	4.24	2	96.50	9.19	2	-4.50	4.95	1	97.00	0	1	90.00	0	1	-7.00	0
Other	53	95.64	5.51	53	93.79	7.91	53	-1.85	6.12	28	97.32	6.22	28	94.29	8.62	28	-3.04	6.50
All																		

TABLE 10  
PROC REPORT OUTPUT OF WORK.NEWMBP2  
SUMMARY REPORT

Race	Gender																	
	Baseline Diastolic BP			Final Diastolic BP			Change in Diastolic BP			Baseline Diastolic BP			Final Diastolic BP			Change in Diastolic BP		
No.	Mean	Std. Dev.	No.	Mean	Std. Dev.	No.	Mean	Std. Dev.	No.	Mean	Std. Dev.	No.	Mean	Std. Dev.	No.	Mean	Std. Dev.	
White	44	94.70	5.52	44	92.73	8.00	44	-1.98	6.32	23	97.13	6.57	23	93.91	9.23	23	-3.22	6.79
Black	7	100.00	1.91	7	99.71	4.27	7	-0.29	5.35	4	98.50	5.45	4	97.50	4.80	4	-1.00	5.60
Other	2	101.00	4.24	2	96.50	9.19	2	-4.50	4.95	1	97.00	0	1	90.00	0	1	-7.00	0
All	53	95.64	5.51	53	93.79	7.91	53	-1.85	6.12	28	97.32	6.22	28	94.29	8.62	28	-3.04	6.50